Wiskundige modellering in de ingenieurswetenschappen: Bordoefeningenles 2

Oefening 1

```
> restart: with(plots):with(LinearAlgebra):with(plottools):
  oorsprong := <0,0,0>:
```

Constructie matrix A en vector y:

```
> K1 := <1,2,3>;
  K2 := <1,4,9>;
  A  := <K1|K2>;
  y  := <10.1,7.4,-5.2>;
```

$$K1 := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$K2 := \begin{bmatrix} 1 \\ 4 \\ 9 \end{bmatrix}$$

$$A := \begin{bmatrix} 1 & 1 \\ 2 & 4 \\ 3 & 9 \end{bmatrix}$$

$$y := \begin{bmatrix} 10.1 \\ 7.4 \\ -5.2 \end{bmatrix} \tag{1.1}$$

Heeft dit stelsel een oplossing?

```
> solve(A.(<v0, -g/2>)=y, {v0,g});
```

```
> Determinant(<K1|K2|y>);
```

$$5.8 \tag{1.2}$$

We bepalen de kleinste kwadraten benadering en fit:

De onbekenden x bepalen kan op 2 manieren:

- met een stelsel (meest efficiënt)

```
> solve((A^%T.A).(<v0,-g/2>)=A^%T.y,{v0,g});
```

$$\{g = 11.42631579,\ v0 = 15.35526316\} \tag{1.3}$$

- met behulp van de matrix inverse

```
> x    := MatrixInverse(A^%T.A).A^%T.y;
```

```
v0    := x[1];
g     := -2*x[2];
```

$$x := \begin{bmatrix} 15.3552631578947 \\ -5.71315789473684 \end{bmatrix}$$

$$v0 := 15.3552631578947$$

$$g := 11.4263157894737 \tag{1.4}$$

De kleinste kwadraten benadering vinden we als

```
> y_kk  := A.x;
  y_com := y-y_kk:
```

$$y\_kk := \begin{bmatrix} 9.64210526315790 \\ 7.85789473684211 \\ -5.35263157894736 \end{bmatrix} \tag{1.5}$$

Visualisatie kolomruimte K(A) + nulruimte N(A^T)

```
> KA1        := plot3d(h*K1+v*K2, h=-8..8,v=-9..9,
             color=green, numpoints=20,style=surface, axes=
  normal):
  KA2        := implicitplot3d((<xx,yy,zz>).y_com=0, xx=-5..5,yy=
  -5..5,zz=-5..5, color=green, numpoints=20,style=surface, axes=
  normal):

  NAT        := plot3d(t*y_com,t=-5..5, linestyle="dot"):

  K1_lijn    := line(oorsprong,K1, color=blue):
  K2_lijn    := line(oorsprong,K2, color=blue):

  y_lijn     := line(oorsprong,y, color=red):
  y_kk_lijn  := line(oorsprong,y_kk, color=purple):
  y_com_lijn := line(oorsprong,2*y_com, color=purple):


  display(KA2, NAT, K1_lijn, K2_lijn, y_lijn, y_kk_lijn,
  y_com_lijn, scaling=constrained);
```



Ten slotte visualiseren we de gevonden fit oplossingen met de datapunten:
We beginnen met de gemeten en geprojecteerde data punten te visualiseren:

```
> data_points    := pointplot(K1, y,    color=blue, legend=
  "gemeten data-punten"):
```

Daarna plotten we het traject (=de gevonden fit):

```
> traject := t->v0*t-g/2*t**2:
```

```
> traject_plot := plot(traject(t),t=0..4.5,    color=red, legend=
  "gevonden fit"):
```

```
> display(data_points,  traject_plot,view=[0..3.5,-7.5..11]);
```



Oefening 5

```
> restart: with(LinearAlgebra):with(plots):
```

Intuïtie:

```
> shadebetween(cos(2*t)*sin(t),0, t=-Pi..Pi);
```



(1) Orthonormale basis:

```
> assume(k,integer):
  assume(l,integer);
```

```
> inp:=(f,g)->int(conjugate(f)*g,t=-Pi..Pi);
```

$$inp := (f, g) \mapsto \int_{-\pi}^{\pi} \bar{f} \cdot g \, dt \tag{2.1}$$

```
> inp(1,1);
  int(1*conjugate(1),t=-Pi..Pi);
```

$$2\pi$$

$$2\pi \tag{2.2}$$

```
> inp(cos(k*t),1);
  inp(cos(k*t),cos(l*t));
  inp(cos(k*t),sin(l*t));
  inp(cos(k*t),sin(k*t));
  inp(sin(k*t),sin(l*t));
  inp(sin(k*t),1);
```

$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0 \qquad (2.3)$$

```
> inp(1,1);
  inp(cos(k*t),cos(k*t));
  inp(sin(k*t),sin(k*t));
```

$$2\pi$$
$$\pi$$
$$\pi \qquad (2.4)$$

De nieuwe basisfuncties:

```
> u0    := 1/sqrt(2*Pi);
  u[k]  := cos(k*t)/sqrt(Pi);
  ut[k] := sin(k*t)/sqrt(Pi);
```

$$u0 := \frac{\sqrt{2}}{2\sqrt{\pi}}$$
$$u_{k\sim} := \frac{\cos(k\sim t)}{\sqrt{\pi}}$$
$$ut_{k\sim} := \frac{\sin(k\sim t)}{\sqrt{\pi}} \qquad (2.5)$$

(2) De projectie van $f(t) = t^2$:

```
> const   := inp(u0, t^2) * u0;
```

$$const := \frac{\pi^2}{3} \qquad (2.6)$$

```
> som_cos :=  Sum( inp(u[k] , t^2) *  u[k], k = 1..n);
  som_sin :=  Sum( inp(ut[k], t^2) * ut[k], k = 1..n);
```

$$som\_cos := \sum_{k\sim=1}^{n} \frac{4\,(-1)^{k\sim}\cos(k\sim t)}{k\sim^2}$$
$$som\_sin := \sum_{k\sim=1}^{n} 0 \qquad (2.7)$$

```
> f0 := const  + som_cos;
```

$$f0 := \frac{\pi^2}{3} + \left(\sum_{k\sim=1}^{n} \frac{4\,(-1)^{k\sim}\cos(k\sim t)}{k\sim^2}\right) \qquad (2.8)$$

```
> plot2  := plot(subs(n=2,f0),  t=-Pi..Pi   ,legend='n=2'):
  plot10 := plot(subs(n=10,f0), t=-Pi..Pi   ,legend='n=10'):
  plot50 := plot(subs(n=50,f0), t=-Pi..Pi   ,legend='n=50'):
```

```
> plot_exact := plot(t^2,            t=-Pi..Pi   ,legend='t^2',
  linestyle="dot", thickness=5):
> display(plot_exact, plot2, plot10, plot50);
```



Enkele inproducten (voor berekening afstand):

```
> inp(t^2,t^2);
  inp(1,t^2);
  inp(cos(k*t),t^2);
  inp(1,1);
```

```
inp(cos(k*t),cos(k*t));
```

$$\frac{2\pi^5}{5}$$
$$\frac{2\pi^3}{3}$$
$$\frac{4\pi(-1)^{k\sim}}{k\sim^2}$$
$$2\pi$$
$$\pi \tag{2.9}$$

(3) Nu nog de afstand, die we berekenen met behulp van de van het inproduct afgeleide norm:

```
> Int(t^4-2*f0*t^2+Pi^4/9+Sum(16/k^4*cos(k*t)^2,k=1..n),t=-Pi..Pi);
```

$$\int_{-\pi}^{\pi}\left(t^4-2\left(\frac{\pi^2}{3}+\left(\sum_{k\sim=1}^{n}\frac{4(-1)^{k\sim}\cos(k\sim t)}{k\sim^2}\right)\right)t^2+\frac{\pi^4}{9}+\left(\sum_{k\sim=1}^{n}\frac{16\cos(k\sim t)^2}{k\sim^4}\right)\right)\mathrm{d}t \tag{2.10}$$

```
> inp(t^2,t^2)-2*int(f0*t^2,t=-Pi..Pi)+int(f0*f0,t=-Pi..Pi);
```

$$\frac{2\pi^5}{5}-2\pi\left(\frac{\pi^4}{9}+\left(\sum_{k\sim=1}^{n}\frac{4(-1)^{k\sim}\left(\pi^2\sin(k\sim\pi)\,k\sim^2+2\pi\cos(k\sim\pi)\,k\sim-2\sin(k\sim\pi)\right)}{k\sim^5\,\pi}\right)\right) \tag{2.11}$$
$$-\frac{1}{9}\left(2\sqrt{\pi}\left(\pi^{9/2}+144\left(\sum_{k\sim=1}^{n}\frac{1}{4\sqrt{\pi}\,k\sim^5}\left((-1)^{k\sim}\left(\pi^2\sin(k\sim\pi)\,k\sim^2+2\pi\cos(k\sim\pi)\,k\sim-2\sin(k\sim\pi)\right)\right)\right)\right)\right)+\pi\left(\frac{\pi^4}{9}+\left(\sum_{k\sim=1}^{n}\frac{4(-1)^{k\sim}\sin(k\sim\pi)\,\pi}{3\,k\sim^3}\right)\right.$$
$$+\left(\sum_{k\sim=1}^{n}\frac{1}{3\,k\sim^3}\left(3\left(\sum_{k\sim=1}^{n}\frac{1}{\pi\,k\sim^5}\left(4(-1)^{2k\sim}\cos(k\sim\pi)\left(2\pi\cos(k\sim\pi)\right.\right.\right.\right.\right.$$
$$\left.k\sim+\sin(2k\sim\pi)\cos(k\sim\pi)-\cos(2k\sim\pi)\sin(k\sim\pi)+\sin(k\sim\pi)\right)\Big)\Big)\,k\sim^3$$
$$+3\left(\sum_{k\sim=1}^{n}\left(-\frac{1}{k\sim^5\,\pi}\left(4(-1)^{2k\sim}\sin(k\sim\pi)\left(-2\pi\sin(k\sim\pi)\,k\sim\right.\right.\right.\right.$$

$$+\sin(2k\sim\pi)\sin(k\sim\pi)+\cos(k\sim\pi)\cos(2k\sim\pi)-\cos(k\sim\pi)))\Bigg)\Bigg)k\sim^3$$

$$+4(-1)^{k\sim}\pi\sin(k\sim\pi)\Bigg)\Bigg)\Bigg)\Bigg)+\pi\left(\frac{\pi^4}{9}+\frac{4\pi\left(\sum_{k\sim=1}^{n}\frac{(-1)^{k\sim}\sin(k\sim\pi)}{k\sim^3}\right)}{3}\right.$$

$$+\left(\sum_{k\sim=1}^{n}\frac{1}{3k\sim^3}\left(3\left(\sum_{k\sim=1}^{n}\frac{4(-1)^{2k\sim}(2k\sim\pi+\sin(2k\sim\pi))}{k\sim^5\pi}\right)k\sim^3\right.\right.$$

$$\left.\left.\left.+4(-1)^{k\sim}\pi\sin(k\sim\pi)\right)\right)\right)$$

```
> simplify(%);
```

$$\frac{8\pi\left(\pi^4+90\left(\sum_{k\sim=1}^{n}\sum_{k\sim=1}^{n}\frac{1}{k\sim^4}\right)-180\left(\sum_{k\sim=1}^{n}\frac{1}{k\sim^4}\right)\right)}{45} \tag{2.12}$$

```
> expand(%);
```

$$\frac{8\pi^5}{45}+16\pi\left(\sum_{k\sim=1}^{n}\frac{1}{k\sim^4}\right)n-32\pi\left(\sum_{k\sim=1}^{n}\frac{1}{k\sim^4}\right) \tag{2.13}$$

Vereenvoudigd geeft dit

```
> normsquared:=simplify(int(t^4-2*f0*t^2+Pi^4/9+Sum(16/k^4*cos(k*t)
  ^2,k=1..n),t=-Pi..Pi));
```

$$normsquared := \frac{8\pi\left(\pi^4-90\left(\sum_{k\sim=1}^{n}\frac{1}{k\sim^4}\right)\right)}{45} \tag{2.14}$$

De limiet als n naar oneindig nadert, is:

```
> (Pi^4-90*Sum(1/(k^4),k = 1 .. infinity)) = 8/45*Pi*(Pi^4-90*sum
  (1/(k^4),k = 1 .. infinity));
```

$$\pi^4-90\left(\sum_{k\sim=1}^{\infty}\frac{1}{k\sim^4}\right)=0 \tag{2.15}$$

Oefening 2

```
> restart:with(LinearAlgebra):
> A := <<0|-1|3|0>,<1|0|0|1>,<0|0|3|-1>,<0|0|1|1>>;
```

$$A := \begin{bmatrix} 0 & -1 & 3 & 0 \\ 1 & 0 & 0 & 1 \\ 0 & 0 & 3 & -1 \\ 0 & 0 & 1 & 1 \end{bmatrix} \tag{3.1}$$

```
> J, Q := JordanForm(A, output=[J, Q]);
```

```
Error, invalid input: LinearAlgebra:-JordanForm expects value for keyword parameter output to be of type {list(identical(J,Q)), identical(J,Q)}, but received [Matrix(4, 4, [[-I,0,0,0],[0,I,0,0],[0,0,2,1],[0,0,0,2]]), Matrix(4, 4, [[-1/2,-1/2,1,1],[-1/2*I,1/2*I,1,0],[0,0,1,1],[0,0,1,0]])]
```

```
> J, Q := JordanForm(A, output=['J', 'Q']);
```

$$J, Q := \begin{bmatrix} -I & 0 & 0 & 0 \\ 0 & I & 0 & 0 \\ 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} -\frac{1}{2} & -\frac{1}{2} & 1 & 1 \\ -\frac{I}{2} & \frac{I}{2} & 1 & 0 \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 1 & 0 \end{bmatrix} \tag{3.2}$$

```
> Q.J.MatrixInverse(Q);
```

$$\begin{bmatrix} 0 & -1 & 3 & 0 \\ 1 & 0 & 0 & 1 \\ 0 & 0 & 3 & -1 \\ 0 & 0 & 1 & 1 \end{bmatrix} \tag{3.3}$$

Oefening 3

```
> restart:
> eq1 := alpha * v1v1 + beta * v1v2 + gamma * v1v3 = 0;
  eq2 := alpha * v1v2 + beta * v2v2 + gamma * v2v3 = 0;
```

$$eq1 := \alpha\, v1v1 + \beta\, v1v2 + \gamma\, v1v3 = 0$$

$$eq2 := \alpha\, v1v2 + \beta\, v2v2 + \gamma\, v2v3 = 0 \tag{4.1}$$

```
> solve({eq1,eq2}, {alpha, beta});
```

$$\left\{\alpha = \frac{\gamma\,(v1v2\, v2v3 - v1v3\, v2v2)}{v2v2\, v1v1 - v1v2^2}, \beta = -\frac{\gamma\,(v1v1\, v2v3 - v1v2\, v1v3)}{v2v2\, v1v1 - v1v2^2}\right\} \tag{4.2}$$

```
> a := (v1v2*v2v3 - v1v3*v2v2)/(v1v1*v2v2 - v1v2^2);
  b := -1*(v1v1*v2v3 - v1v2*v1v3)/(v1v1*v2v2 - v1v2^2);
```

$$a := \frac{v1v2\, v2v3 - v1v3\, v2v2}{v2v2\, v1v1 - v1v2^2}$$

$$b := -\frac{v1v1\, v2v3 - v1v2\, v1v3}{v2v2\, v1v1 - v1v2^2} \tag{4.3}$$

Oefening 4

```
> restart: with(LinearAlgebra):
> e1 := <1,0>;
  e2 := <0,1>;
```

$$e1 := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$e2 := \begin{bmatrix} 0 \\ 1 \end{bmatrix} \tag{5.1}$$

```
> A := 1/sqrt(2) * (e1.e2^%T + e2.e1^%T);
```

$$A := \begin{bmatrix} 0 & \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & 0 \end{bmatrix} \tag{5.2}$$

```
> A=A^%T;
  A^2=A;
```

$$\begin{bmatrix} 0 & \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & 0 \end{bmatrix} = \begin{bmatrix} 0 & \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & 0 \end{bmatrix}$$

$$\begin{bmatrix} \frac{1}{2} & 0 \\ 0 & \frac{1}{2} \end{bmatrix} = \begin{bmatrix} 0 & \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & 0 \end{bmatrix} \tag{5.3}$$