Wiskundige modellering in de ingenieurswetenschappen: Werkcollege 3

Oefening 1

```
> restart:with(plots):with(LinearAlgebra):
```

We definiëren de (6x6)-matrix A als functie van Ex, Ey en B

```
> A := (Ex, Ey, B)-><<0|0|B|0|Ex>,<1|0|0|0|0>,<-B|0|0|0|Ey>,
  <0|0|1|0|0>,<0|0|0|0|0>>;
  A(Ex,Ey,B);
```

$$A := (Ex, Ey, B) \mapsto \langle\langle 0|0|B|0|Ex\rangle, \langle 1|0|0|0|0\rangle, \langle -B|0|0|0|Ey\rangle, \langle 0|0|1|0|0\rangle, \langle 0|0|0|0|0\rangle\rangle$$

$$\begin{bmatrix} 0 & 0 & B & 0 & Ex \\ 1 & 0 & 0 & 0 & 0 \\ -B & 0 & 0 & 0 & Ey \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad (1.1)$$

Ook de beginwaarden steken we in een vector(functie)

```
> X0 := (x0,y0,vx0,vy0) -> <vx0,x0,vy0,y0,1>:
  X0(x0,y0,vx0,vy0);
```

$$\begin{bmatrix} vx0 \\ x0 \\ vy0 \\ y0 \\ 1 \end{bmatrix} \quad (1.2)$$

De algemene oplossing wordt gevonden als

```
> sol   := MatrixExponential(t*A(Ex,Ey,B)).X0(x0,y0,vx0,vy0);
  x_sol := sol[2];
  y_sol := sol[4];
```

$$sol := \left[\left[\cos(tB)\,vx0 + \sin(tB)\,vy0 + \frac{Ex\sin(tB) - Ey\cos(tB) + Ey}{B}\right],\right.$$

$$\left[\frac{\sin(tB)\,vx0}{B} + x0 + \frac{(-\cos(tB)+1)\,vy0}{B} + \frac{B\,Ey\,t - Ex\cos(tB) - Ey\sin(tB) + Ex}{B^2}\right],$$

$$\left[-\sin(tB)\,vx0 + \cos(tB)\,vy0 + \frac{Ex\cos(tB) + Ey\sin(tB) - Ex}{B}\right],$$

$$\left[\frac{(\cos(t B) - 1)\, vx0}{B} + \frac{\sin(t B)\, vy0}{B} + y0 + \frac{-B\, Ex\, t + Ex \sin(t B) - Ey \cos(t B) + Ey}{B^2}\right],$$

$$\left[1\right]\Big]$$

$$x\_sol := \frac{\sin(t B)\, vx0}{B} + x0 + \frac{(-\cos(t B) + 1)\, vy0}{B} + \frac{B\, Ey\, t - Ex \cos(t B) - Ey \sin(t B) + Ex}{B^2}$$

$$y\_sol := \frac{(\cos(t B) - 1)\, vx0}{B} + \frac{\sin(t B)\, vy0}{B} + y0 + \frac{-B\, Ex\, t + Ex \sin(t B) - Ey \cos(t B) + Ey}{B^2} \tag{1.3}$$

Kijken we nu naar de concrete beginvoorwaarden voor (Ex,Ey) = (0,0), B=10

```
> sol_E0 := MatrixExponential(t*A(0,0,10)).X0(0,0,2,-1):
  x_solE0 := sol_E0[2]:
  y_solE0 := sol_E0[4]:
> plot([x_solE0, y_solE0,t=0..5]);
```



Analoog voor (Ex,Ey)=(-1,2)

```
> sol_E := MatrixExponential(t*A(-1,2,10)).X0(0,0,2,-1):
  x_solE := sol_E[2]:
  y_solE := sol_E[4]:
> plot([x_solE, y_solE,t=0..5]);
```



Het qualitatieve verschil is te zien in de Jordanvorm van de matrix A: indien E verschillend is van 0 verschijnen er Jordan-blokken, wat zal zorgen voor een verschillend gedrag wanneer men e exponentiële exp(t*A) neemt

```
> JE0 := JordanForm(A(0,0,B));
  JE  := JordanForm(A(Ex,Ey,B));
```

$$JE0 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & -\mathrm{I}B & 0 & 0 & 0 \\ 0 & 0 & \mathrm{I}B & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$JE := \begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -\mathrm{I}B & 0 & 0 \\ 0 & 0 & 0 & \mathrm{I}B & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{1.4}$$

▶ Oefening 3

▶ Oefening 2