

Werkcollege 8

► Oefening 1

▼ Oefening 2

```
[> restart;
```

```
Vloeistoftank met een capaciteit van  $V$  liter bevat aanvankelijk  $W$  liter ( $W < V$ ) water, waarin  $Z$  kilogram zout is opgelost.
```

```
FASE 1: IN  $s$  l/min (zuiver water) // UIT  $r$  l/min (homogeen), ( $r < s$ ) // STOP als tank vol is (tijd  $t_1$ ).
```

```
FASE 2: IN  $r$  l/min (zout water,  $k$  kg/l,  $k$  groter dan einde fase 1) // UIT  $r$  l/min
```

▼ Deel (a)

Bepaal op elk ogenblik van de eerste fase de hoeveelheid zout in de tank.

```
[> vloe1:= D(v1)(t)=s-r;  
[> sysv1:= {vloe1,v1(0)=W};  
[> dsolve(sysv1);  
[> v1:= unapply(rhs(%),t);  
[> t1:= solve(v1(t)=V,t);  
[> zout1:= D(z1)(t)=-r*z1(t)/v1(t);  
[> sysz1:= {zout1, z1(0)=Z};  
[> dsolve(sysz1);  
[> z1:=simplify(unapply(rhs(%),t));  
[> kappa:=r/(r-s);  
[Hoeveelheid zout op het einde van fase 1  
[> Z1:=simplify(z1(t1));
```

▼ Deel (b)

Bepaal op elk ogenblik van de tweede fase de hoeveelheid zout in de tank.

```
[> v2:=t->V;  
[> zout2:= D(z2)(t) = -r*z2(t)/v2(t) + k*r;  
[> sysz2:= {zout2, z2(t1)=Z1};  
[> dsolve(sysz2);  
[> z2:= simplify(unapply(rhs(%),t));
```

▼ Deel (c)

Bepaal op elk tijdstip de hoeveelheid vloeistof $v(t)$ en de hoeveelheid zout $z(t)$ in de tank.

```
[> v:=t->piecewise(t<0,0, 0<=t and t<=t1,v1(t), v2(t));  
[> z:=t->piecewise(t<0,0, 0<=t and t<=t1,z1(t), z2(t));
```

▼ Deel (d)

Voorspel de waarde van $\lim_{t \rightarrow \infty} (z(t))$.

```
[> limit(z2(t),t=+infinity);
```

▼ Deel (e)

Hoelang moet de tweede fase duren om de hoeveelheid zout die in de tank aanwezig is na het beëindigen van de eerste fase te verdrievoudigen?

Zoek tijdstip $t_2 > t_1$ waarvoor $z(t_2) = Nz(t_1)$ met $N = 3$. De gevraagde duur is dan het verschil in tijd $t_2 - t_1$.

```
[> t2:= solve(z2(t)=3*Z1,t);  
[> duur:=simplify(t2-t1);
```

▼ Deel (f)

Visualiseer het volledige verloop (
 $V = 500, W = 100, Z = 10, s = 15, r = 5, k = 0.2$)

```
[> V:=500;  
[> W:=100;  
[> Z:=10;  
[> s:=15;  
[> r:=5;  
[> k:=0.2;  
[> v(t);  
[> z(t);  
[> evalf(3*Z1) < evalf(k*V);  
[> evalf(Z1)<evalf(k*V);  
[> evalf(duur);  
[> evalf(t1);  
[> plot([v(t),z(t)],t=0..200, color=[blue,red], legend=  
[> ["vloeistof","zout"]);  
[> limit(z(t),t=+infinity);  
[> k*V;
```

