Wiskundige modellering in de ingenieurswetenschappen: Werkcollege 10

Oefening 2.1 (g(x)=-x^2+1.5x+1.5)

```
> restart:with(plots):
```

Invoeren gegevens:

```
> L   := 2;
  T0  := 1.5;
  a   := 0.3;
```

$$L := 2$$

$$T0 := 1.5$$

$$a := 0.3 \qquad (1.1)$$

```
> g := x -> -x**2+1.5*x+1.5;
```

$$g := x \mapsto -x^2 + 1.5 \cdot x + 1.5 \qquad (1.2)$$

```
> plot(g(x), x=0..L, y=0..2.2,legend="u(x,t=0)")
```



Coefficienten s_n

```
> s := n -> 2/L * int(sin(((2*n+1)*Pi*x)/(2*L))*(g(x)-T0), x=0..L);
```

$$s := n \mapsto \frac{2 \cdot \left(\int_0^L \sin\left(\frac{(2 \cdot n + 1) \cdot \pi \cdot x}{2 \cdot L} \right) \cdot (g(x) - T0) \, dx \right)}{L} \tag{1.3}$$

```
> (simplify(s(n) assuming(n,posint)));
```

$$\frac{4.128196410 + (-8.105694697\, n - 4.052847348)\, (-1)^n}{(2.\, n + 1.)^3} \tag{1.4}$$

Normal Modes

```
> nmode := (n,x,t) -> exp(  -((2*n+1)*Pi*a/(2*L))**2 * t) * sin((
  (2*n+1)*Pi*x)/(2*L))
```

$$nmode := (n, x, t) \mapsto e^{-\frac{(2 \cdot n + 1)^2 \cdot \pi^2 \cdot a^2 \cdot t}{4 \cdot L^2}} \cdot \sin\left(\frac{(2 \cdot n + 1) \cdot \pi \cdot x}{2 \cdot L} \right) \tag{1.5}$$

Homogene oplossing

```
> uh := (x,t,N) -> sum(s(n)*nmode(n,x,t), n=0..N);
```

$$uh := (x, t, N) \mapsto \sum_{n=0}^{N} s(n) \cdot nmode(n, x, t) \tag{1.6}$$

Tijdsonafhankelijke inhomogene oplossing

```
> ut := x-> T0;
```

$$ut := x \mapsto T0 \tag{1.7}$$

Algemene oplossing

```
> u := (x,t,N) -> ut(x) + uh(x,t,N):
```

Visualisatie

```
> N := 100;
```

$$N := 100 \tag{1.8}$$

```
> animate(plot, [u(x,t,N), x = 0 .. L, y=0..2.2], t = 0 .. 10,
  frames=50);
```




```
> dens := (t,x,y,N)-> u(x,t,N):
> animate(densityplot, [dens(t,x,y,50), x=0..L, y=0..1,
  restricttoranges=true], t = 0 .. 20, frames=50);
```

Oefening 2.2 (g(x)= stapfunctie)

```
> restart:with(plots):
```

Invoeren gegevens:

```
> L    := 2;
  T0   := 1.5;
  a    := 0.3;
```

$$L := 2$$
$$T0 := 1.5$$
$$a := 0.3 \qquad (2.1)$$

```
> g := x-> piecewise(x<1,1.5, x>1,0.5);
```

$$g := x \mapsto \begin{cases} 1.5 & x < 1 \\ 0.5 & 1 < x \end{cases} \qquad (2.2)$$

```
> plot(g(x), x=0..L, y=0..2.2,legend="u(x,t=0)")
```



Coefficienten s_n

```
> s := n -> 2/L * int(sin(((2*n+1)*Pi*x)/(2*L))*(g(x)-T0), x=0..L);
```

$$s := n \mapsto \frac{2\cdot\left(\int_0^L \sin\left(\frac{(2\cdot n+1)\cdot\pi\cdot x}{2\cdot L}\right)\cdot(g(x)-T0)\,\mathrm{d}x\right)}{L} \tag{2.3}$$

```
> (simplify(s(n) assuming(n,posint)));
```

$$-\frac{1.273239545\cos(1.570796327\,n+0.7853981634)}{2.\,n+1.} \tag{2.4}$$

Normal Modes

```
> nmode := (n,x,t) -> exp(  -((2*n+1)*Pi*a/(2*L))**2 * t) * sin((
  (2*n+1)*Pi*x)/(2*L))
```

$$nmode := (n, x, t) \mapsto \mathrm{e}^{-\frac{(2\cdot n+1)^2\cdot\pi^2\cdot a^2\cdot t}{4\cdot L^2}}\cdot\sin\left(\frac{(2\cdot n+1)\cdot\pi\cdot x}{2\cdot L}\right) \tag{2.5}$$

Homogene oplossing

```
> uh := (x,t,N) -> sum(s(n)*nmode(n,x,t), n=0..N);
```

(2.6)

$$uh := (x, t, N) \mapsto \sum_{n=0}^{N} s(n) \cdot nmode(n, x, t) \tag{2.6}$$

Tijdsonafhankelijke inhomogene oplossing

```
> ut := x-> T0;
```

$$ut := x \mapsto T0 \tag{2.7}$$

Algemene oplossing

```
> u := (x,t,N) -> ut(x) + uh(x,t,N):
```

Visualisatie

```
> N := 100;
```

$$N := 100 \tag{2.8}$$

```
> animate(plot, [u(x,t,N), x = 0 .. L, y=0..2.2], t = 0 .. 10,
  frames=50);
```




```
> dens := (t,x,y,N)-> u(x,t,N):
> animate(densityplot, [dens(t,x,y,50), x=0..L, y=0..1,
  restricttoranges=true], t = 0 .. 20, frames=50);
```

0Oefening 1

```
> restart:with(plots):
```

Oplossing differentiaalvergelijking: karakteristieke vergelijking

```
> solve(z**2+z+(n**2*Pi**2)=0, z) assuming(n,posint);
```

$$-\frac{1}{2}+\frac{I\sqrt{4\pi^2 n^2-1}}{2},\ -\frac{1}{2}-\frac{I\sqrt{4\pi^2 n^2-1}}{2} \quad (3.1)$$

Beginvoorwaarden

```
> f := x -> -3*x*(x-1);
  g := x -> sin(5*Pi*x)*2*x;
```

$$f := x \mapsto -3\cdot x\cdot (x-1)$$
$$g := x \mapsto 2\cdot \sin(5\cdot\pi\cdot x)\cdot x \quad (3.2)$$

```
> plot({f(x), g(x)},x=0..1);
```




```
> s := n -> 2*int(sin(n*Pi*x)*f(x),x=0..1):
  c := n -> 4/sqrt(4*n**2*Pi**2-1)*int(sin(n*Pi*x)*(g(x)+1/2*f(x)),
  x=0..1):
> u := (x,t,N) -> sum((s(n)*cos(sqrt(4*n**2*Pi**2-1)/2 * t) + c(n)*
  sin(sqrt(4*n**2*Pi**2-1)/2 * t))*exp(-t/2)*sin(n*Pi*x), n=1..N):
> animate(plot, [{u(x,tt,80),f(x)}, x=0..1], tt=0..15, frames=80);
```




```
> plot0 := plot(u(x,0,80), x=0..1, color=red, legend="t=0"):
  plot1 := plot(u(x,0.2,80), x=0..1, color=blue, legend="t=0.2"):
  plot2 := plot(u(x,0.5,80), x=0..1, color=black, legend="t=0.5"):
  plot3 := plot(u(x,0.95,80), x=0..1, color=purple, legend="t=
  0.95"):
> display(plot0, plot1, plot2, plot3);
```



0.6
0.4
0.2
0
−0.2
0.4
0.2
0.4
0.6
0.8
1
x
t=0
t=0.2
t=0.5
t=0.95